

Caractérisation d'un amplificateur haute fréquence

J'invite le lecteur à consulter le site pour des informations complémentaires.

Page d'accueil du site Internet :


[page d'accueil](#)

d'autres pdf, sur différents sujets :

[liste des PDE](#)

Le schéma analysé dans ce document a pour source :

https://www.artedas.fr/documentsPDF/Guide_evaluation/Guide_OrcadDemo166_DVD_Artedas.pdf.



Décrit en page 55, ce circuit sert de base pour exploiter l'analyse de sensibilité, qui est une fonctionnalité récente associée à Pspice. On se propose dans ces pages d'analyser cet amplificateur hautes fréquences, et de vérifier ses paramètres et performances.

Ci-dessous la netliste reconstruite à partir du schéma :

Amplificateur haute fréquence

* fichier ampli_HF.cir

.LIB EVAL.LIB ; pour diode

* circuit

V1 vcc 0 DC=15

D1 vcc 1 D1N4148

R2 1 B4 3k

R3 B4 0 6.8k

Q4 C4 B4 E4 Q2N3905

C1 C4 0 10n

R1 C4 B3 24k

R6 vcc E4 470

R7 E4 C3 270

C4 E4 0 10u

Q3 C3 B3 E3 Q2N5179

R8 E3 0 3.3

C7 C3 i 0.47u

R4 i B3 470

V2 In 0 sin (0.5m 1Meg) AC 1

R5 In Inn 50

C6 Inn B3 0.47u

C3 C3 Load 0.47u

R9 Load 0 50

.OP

.AC DEC 1000 1k 1G

.TRAN 0.1ns 10u 0n 0.1ns

.probe

.MODEL Q2N3905 PNP(Is=1.41f Xti=3 Eg=1.11 Vaf=18.7 Bf=90.35 Ne=1.5 Ise=0
+ Ikf=80m Xtb=1.5 Br=5.502 Nc=2 Isc=0 Ikr=0 Rc=2.5 Cjc=9.728p
+ Mjc=.5776 Vjc=.75 Fc=.5 Cje=8.063p Mje=.3677 Vje=.75 Tr=35.05n
+ Tf=179.3p Itf=.4 Vtf=4 Xtf=6 Rb=10)

.model Q2N5179 NPN(Is=69.28E-18 Xti=3 Eg=1.11 Vaf=100 Bf=282.1 Ne=1.177
+ Ise=69.28E-18 Ikf=22.03m Xtb=1.5 Br=1.176 Nc=2 Isc=0 Ikr=0 Rc=4
+ Cjc=893.1f Mjc=.3017 Vjc=.75 Fc=.5 Cje=939.8f Mje=.3453 Vje=.75
+ Tr=1.588n Tf=141.1p Itf=.27 Vtf=10 Xtf=30 Rb=10)

.end

Netlist, prête à simuler

1) Polarisation

Hypothèses :

$I_{B4} \ll$ courant dans le pont D_1, R_2, R_3 .

$I_{E4} \ll$ courant dans $R_6, R_7 = I_{C3}$.

Le transistor Q_4 travaille à V_{EB} constant, et donc I_{B4} est constant.

D'où $I_{C4} = \text{constant} = I_{B3}$.

Le transistor Q_3 a donc I_C constant. La résistance d'émetteur R_8 apporte une stabilisation thermique.

On vérifie la polarisation : le choix des valeurs numériques de R_2, R_3, R_6 permet avoir les 2 transistors dans le mode normal.

On a, après simulation : $I_{D1} = 1,47$ mA. La première hypothèse est vérifiée.

On a $I_{E4} = 0,128$ mA. La deuxième hypothèse est vérifiée.

Reprenons les équations, et donnons une représentation par le formalisme des schémas blocs :

$$V_{B4} = (V_{CC} - V_{D1}) R_3 / (R_2 + R_3)$$


$$V_{EB4} = V_{E4} - V_{B4}$$

$$I_{B4} = f(V_{EB4}) \quad I_{C4} = \beta_4 I_{B4}$$

$$I_{C3} = \beta_3 I_{B3} \quad I_{C4} = I_{B3}$$

$$V_{E4} = V_{CC} - R_6 I_{C3}$$

$$V_{CE3} = V_{CC} - (R_6 + R_7 + R_8) I_{C3}$$



NAME	Q4	Q3
MODEL	Q2N3905	Q2N5179
IB	-1.07E-06	1.28E-04
IC	-1.28E-04	9.18E-03
VBE	-6.45E-01	8.49E-01
VBC	6.02E+00	-7.26E+00
VCE	-6.67E+00	8.11E+00
BETADC	1.19E+02	7.16E+01
GM	4.94E-03	2.77E-01
RPI	2.41E+04	2.24E+02
RX	1.00E+01	1.00E+01
RO	1.93E+05	1.17E+04
CBE	1.41E-11	4.52E-11
CBC	2.73E-12	4.38E-13
CJS	0.00E+00	0.00E+00
BETAAC	1.19E+02	6.22E+01
CBX/CBX2	0.00E+00	0.00E+00
FT/FT2	4.69E+07	9.67E+08

On suppose V_{D1} constant. V_{B4} est donc constant. Le transistor Q_4 délivre le courant de base de Q_3 , par la loi (non linéaire) de Shockley. Il y a une contre réaction par I_{C3} qui intervient sur V_{EB4} . On peut remarquer que l'association $Q_4 Q_3$ forme un pseudo Darlington.

La diode D_1 permet, en ayant le même coefficient de température que V_{EBQ4} , de rendre plus robuste le courant de polarisation suite à une variation de température.

2) Amplification et bande passante en petits signaux


Q_3 est monté en émetteur commun, avec résistance d'émetteur, R_8 .

Le choix de $0,47 \mu F$ pour C_6 et C_3 est cohérent avec la fréquence utilisée : à 1 MHz, la réactance est $0,34 \Omega$ soit $\ll 50 \Omega$.

De même C_7 est un court-circuit devant R_4 .

Schéma équivalent du transistor en HF


En hautes fréquences, le transistor Q_3 peut être remplacé par son modèle de Giacoletto.



Le schéma d'étude résultant est donné ci-contre.

C'est trop complexe pour être résolu « à la main », -et c'est l'intérêt de disposer de la simulation-.

Néanmoins, on peut faire des hypothèses simplificatrices et prédéterminer l'amplification moyenne fréquence de ce circuit.



On peut :

- négliger l'influence de r_{CE} : valeur très élevée, et en quasi parallèle à R_4 .
- négliger l'influence de r_0 : valeur très élevée, et en quasi parallèle à $R_7 // R_9$.
- négliger l'influence de $r_{BB'}$: valeur faible (10Ω), devant r_π .

Détermination de l'amplification en moyenne fréquence

En moyenne fréquence, on peut retirer les condensateurs C_{BE} , C_{BC} , de très faible valeur.

Avec les hypothèses simplificatrices précédentes, il vient alors un schéma simplifié :




Schéma équivalent, en moyennes fréquences

- Raisonnons tout d'abord en absence de R_4 :

En moyenne fréquence, la charge est : $R_{EQ} = R_7 // R_9 = 270 // 50 = 42,2 \Omega$.

L'amplification $\frac{V(\text{Load})}{V(\text{Inn})}$ est : $A_v = -\frac{g_m R_{EQ}}{1 + g_m R_8}$.

D'après la polarisation $g_m = 0,277$. Il vient : $A_v = -6,1$.

- Intéressons-nous à la contribution de la résistance R_4 :

On exploite le théorème de Miller. Ici, on prend comme hypothèse que la valeur de A_v est conservée avec la présence de R_4 , ce qui permet de faire un calcul rapide :

R_4 est ramenée :

- en entrée (parallèle à R_1) : $\frac{R_4}{(1 - A_v)} = 470 / (1 + 6,1) = 66,2 \Omega$.

Il y a également à ce point : $(\beta R_8 + r_\pi) = 71,6 \times 3,3 + 224 = 460 \Omega$.

Au total, on a : $66,2 // 460 = 58 \Omega$. La contribution de R_1 ($24 \text{ k}\Omega$) est négligeable.

Cela forme un pont diviseur de tension à l'entrée $\frac{V(\text{Inn})}{V(\text{In})} = \frac{58}{50 + 58} = 0,537$.

- en sortie (parallèle avec R_7) : $\frac{R_4 A_v}{(1 - A_v)} = 470 \cdot 6,1 / 7,1 = 403,8 \Omega$ en sortie (on néglige le rôle de $r_0 = 11,7 \text{ k}\Omega$).


Remarque : la résistance dynamique en $V(\text{Load})$ est $42,2 // 403,8 = 38,2 \Omega$

L'amplification attendue, en moyenne fréquence, $\frac{V(\text{Load})}{V(\text{In})}$ est donc $-0,537 \times 6,1 = -3,27$, soit $10,3 \text{ dB}$

Les auteurs de ce document ont indiqué, page 55 :

« ... un gain maximal de 9.48 et une bande passante à 3 dB de 172 Meg.... »

Vérifions ces valeurs en $V(\text{Load})$ (c'est à dire $V(\text{Load}) / V(\text{In})$) :



Réponse harmonique du schéma proposé

Le plateau est à 9,5 dB, et on lit clairement la fréquence de coupure de 172 MHz :

Measurement	Value
Cutoff_Lowpass_3dB(V(Load))	172.08982meg

On confirme donc les chiffres donnés dans le document d'Orcad.

En affichant V(Load), on peut lire une amplification de toute la chaîne, à 1 MHz : - 2,985.

De même, par $V(\text{Load}) / V(\text{Inn})$, on lit $A_v = - 5,34$. Cette valeur est légèrement inférieure à celle déterminée par notre calcul simplifié (- 6,1 avec des hypothèses simplificatrices).

3) Impédance d'entrée

Maintenant que le schéma est simulé, on peut accéder à d'autres informations, comme l'impédance d'entrée : il suffit d'afficher la grandeur $V(\text{Inn})/I(\text{C6})$ en module et phase :



Module et phase de l'impédance d'entrée présentée par le transistor : contribution de R4

On peut lire, qu'à 1 MHz, l'impédance présentée en Inn est de 63 Ω purement résistif. On avait calculé précédemment 63,7 Ω. C'est en parfaite cohérence.


4) Impédance de sortie

La sortie de l'amplificateur peut être modélisée par une source de Thévenin, formée de sa f.e.m. de sa résistance, qui est la résistance de sortie à déterminer.

On reste dans un domaine de fréquences où les capacités de jonction n'interviennent pas.

Une façon de faire est de relancer le run à une autre valeur de R_9 pour quantifier le pont diviseur de tension formé par l'impédance de sortie et R_9 .

On relance avec $R_9 = 40 \Omega$, et 60Ω .



X Values	1.0000M
V(Load)	2.6375
V(Load)	2.9852
V(Load)	3.2728

On a :
$$V_{Load} = V_{out} \frac{R_9}{R_S + R_9}, \text{ avec } V_{out} = A_v V_{inn}.$$

Par l'essai précédent à $R_9 = 50 \Omega$, on déduit :
$$A_v V_{inn} \frac{50}{R_S + 50} = 2,9852 \text{ V}$$

- Exploitions l'essai à $R_9 = 40 \Omega$:
$$A_v V_{inn} \frac{40}{R_S + 40} = 2,6375 \text{ V}$$

Il vient
$$\frac{2,9852}{2,6375} = 1,13183 = \frac{50}{R_S + 50} \frac{R_S + 40}{40}.$$
 On déduit : $R_S = 55,8 \Omega$


- Exploitions l'essai à $R_9 = 60 \Omega$:
$$A_v \frac{60}{R_S + 60} = 3,2728.$$

Il vient
$$\frac{2,9852}{3,2728} = 0,91212 = \frac{50}{R_S + 50} \frac{R_S + 60}{60}.$$
 On confirme : $R_S = 55,8 \Omega$

5) Fonctionnement à 1 MHz.

- petits signaux


On place une sinusoïde de fréquence 1 MHz, et d'amplitude 5 mV (comme indiqué sur le schéma d'origine).



Pour 10 mV ptp en entrée V(In), on retrouve 5,58 mV ptp en V(Inn). On retrouve quasiment la valeur du pont diviseur calculée.

En sortie V(Load), on mesure, ptp : 29,8 mV en opposition de phase. L'amplification est donc - 2,98, comme déterminé dans la réponse harmonique.

Profitons de ce run pour représenter la droite de charge dynamique en sortie :



Droite de charge dynamique ($V_{C3} = f(I_{C3})$), petits signaux, autour du point de fonctionnement

On a repéré les coordonnées du point de repos : $I_{C3} = 9,18 \text{ mA}$; $V_{C3} = 8,1442 \text{ V}$


La pente de la droite charge dynamique est :

$$\Delta V_{C3} / \Delta I_{C3} = (8,1589 \text{ V} - 8,1292 \text{ V}) / (8,7921 \text{ mA} - 9,573 \text{ mA}) = 29,76 \text{ mV} / 0,780 \text{ mA} = \boxed{38,15 \Omega}$$

C'est la valeur précédemment calculée.

- grands signaux

La notion de grands signaux apparaît quand on quitte le domaine linéaire (ou plutôt linéarisé). Une distorsion apparaît très nettement en sortie pour $V_{in} = 150 \text{ mV}$:



Distorsion sur la tension de sortie à fort niveau

La cause en est une grande dynamique de tension appliquée en V_{BE3} , et donc un courant I_{B3} , puis I_{C3} , non sinusoïdal.

Conclusion

Orcad n'a pas donné tous les renseignements sur ce montage. Cet article a pu chiffrer les impédances d'entrée et de sortie (et de voir qu'on était proche de l'adaptation 50Ω). La notion de distorsion en grands signaux a également été présentée.

On pourrait chercher des modifications de ce montage, comme placer des résistances aux valeurs normalisées pour R_1 et R_2 , modifier la polarisation pour réduire C_{BE} (et C_{BC}), afin d'augmenter la bande passante, d'autant plus que le transistor 2N5179 est donné à $f_T = 1,5 \text{ GHz}$ environ (sous $I_C = 5 \text{ mA}$, $V_{CE} = 6 \text{ V}$). Néanmoins, il faut avouer que pour des fréquences supérieures à 100 MHz , la simulation Pspice, telle qu'elle est présentée ici, n'est pas un outil adapté. Les résultats s'éloignent de la réalité : il faudrait tenir compte d'autres phénomènes intervenant dans le domaine HF, comme le câblage, les inductances parasites, les imperfections des composants, de l'alimentation etc. L'analyse « moyenne fréquence » est insuffisante pour caractériser un tel montage donné pour de la H.F.

articles 1 à 43 : sur le livre

Tableau récapitulatif des articles PDF disponibles sur ce site

n°	titre	lien présentation	lien direct article
	Guide d'installation et d'emploi simplifié	présentation	document PDF
44	Exemples basiques et des exercices...	présentation	document PDF
45	Un exemple de circuit passif	présentation	document PDF
46	Un oscillateur Colpitts	présentation	document PDF
47	Compensation en fréquence des amplificateurs opérationnels	présentation	document PDF
48	Un amplificateur à transistors bipolaires	présentation	document PDF
49	Une bascule D Flip Flop CMOS	présentation	document PDF
50	Une porte XOR à transistors MOS	présentation	document PDF
51	Un VCO à 12 transistors MOS	présentation	document PDF
52	Une PLL à moins de 20 transistors MOS	présentation	document PDF
53	Un oscillateur à résistance négative	présentation	document PDF
54	Une charge électronique	présentation	document PDF
55	Un amplificateur en classe C	présentation	document PDF
56	Le monostable 74 123	présentation	document PDF
57	Un amplificateur en classe D	présentation	document PDF
58	Le transformateur en linéaire	présentation	document PDF
59	La loi d'ohm thermique	présentation	document PDF
60	Le transformateur en non linéaire	présentation	document PDF
61	Robustesse d'un oscillateur en anneau	présentation	document PDF
62	Une alimentation stabilisée	présentation	document PDF
63	Modélisation d'un haut-parleur	présentation	document PDF
64	Un synthétiseur de fréquence	présentation	document PDF
65	Un ampli audio de Sparkfun	présentation	document PDF
66	Simulation logique et analogique	présentation	document PDF
67	Un oscillateur à relaxation	présentation	document PDF
68	Lecteur de TAG RFID 125 kHz	présentation	document PDF
69	Diagramme de l'œil avec Pspice	présentation	document PDF
70	Un amplificateur hautes fréquences	présentation	document PDF
71	Une bizarrerie enfin expliquée...	présentation	document PDF
72	Comprendre le paramétrage de la FFT	présentation	document PDF
73	La relation de Bennett	présentation	document PDF
74	Simuler un circuit à plus de 20 transistors avec PSpice Eval	présentation	document PDF
75	Une horloge biphase sans recouvrement	présentation	document PDF
76	Quelques simulations sur la diode	présentation	document PDF
77	Un ampli classe A, avec transformateur de sortie	présentation	document PDF
78	Des stimuli pour PSpice	présentation	document PDF
79	Simuler le TL431 : zener ajustable	présentation	document PDF
80	Un ADC flash	présentation	document PDF
81	Une chaîne d'acquisition : S&H, ADC, DAC	présentation	document PDF
82	Un amplificateur 50 MHz	présentation	document PDF
83	Un dérivateur non inverseur	présentation	document PDF
84	Un amplificateur bipolaire avec push pull CMOS	présentation	document PDF
85	Rôle des répéteurs logiques dans un circuit intégré	présentation	document PDF
86	Un driver logique CMOS pour charge 50 ohms	présentation	document PDF
87	Des triggers de Schmitt et des applications	présentation	document PDF
88	Un filtre gaussien analogique	présentation	document PDF
89	Un générateur de bruit rose	présentation	document PDF

90	Un anémomètre à fil chaud : simulation comportementale	présentation	document PDF
91	Un oscillateur à pont de Wien stabilisé par CTN	présentation	document PDF
92	L'emballement thermique d'une diode	présentation	document PDF
93	Les puissances dans un amplificateur	présentation	document PDF
94	Asservissement de puissance dans une résistance	présentation	document PDF
95	Asservissement de la puissance émise par une antenne radio	présentation	document PDF
96	Un driver de LED de puissance	présentation	document PDF
97	Exploiter Pspice pour simuler des filtres numériques	présentation	document PDF
98	Un filtre en cosinus surélevé avec Pspice	présentation	document PDF
99	Effet de la température sur un amplificateur en classe A	présentation	document PDF
100	Un amplificateur à transistors JFET et bipolaires	présentation	document PDF
Supplément, hors article :			
mon cours « Electronique pour les communications numériques », polycopié couleur 201 pages en pdf			

[retour à l'écran d'accueil de ce site](#)